

Description

`import fred` imports data from the Federal Reserve Economic Data (FRED) into Stata. `import fred` supports data on FRED as well as historical vintage data on Archival FRED (ALFRED). `freddescribe` and `fredsearch` provide tools to describe series in the database and to search FRED for data based on keywords and tags.

Quick start

Before running any of the commands below, you will need to obtain a FRED key and set it using `set fredkey`.

Import series `code1` and `code2` from FRED

```
import fred code1 code2
```

Import vintage series `code1` and `code2` as available on September 15, 2008, and September 15, 2009, from FRED

```
import fred code1 code2, vintage(2008-9-15 2009-9-15)
```

Display metadata describing series `code1` and `code2`

```
freddescribe code1 code2
```

Search FRED for series matching keywords “investment” and “share” and tagged with “pwt” and “usa”

```
fredsearch investment share, tags(pwt usa)
```

Menu

File > Import > Federal Reserve Economic Data (FRED)

Syntax

Set FRED key

```
set fredkey key [ , permanently ]
```

Import FRED data

```
import fred series_list [ , options ]
```

or

```
import fred, serieslist(filename) [ options ]
```

Describe series

```
freddescribe series_list [ , detail realtime(start end) ]
```

Search series

```
fredsearch keyword_list [ , search_options ]
```

key is a valid API key, which is provided by the St. Louis Federal Reserve and may be obtained from https://research.stlouisfed.org/docs/api/api_key.html.

series_list is a list of FRED codes, for example, FEDFUNDS.

keyword_list is a list of keywords.

options	Description
* <u>serieslist</u> (filename)	specify series IDs using a file
<u>daterange</u> (start end)	restrict to only observations within specified date range
<u>aggregate</u> (frequency[, method])	specify the aggregation level and aggregation type
<u>realtime</u> (start end)	import historical vintages between specified dates
<u>vintage</u> (datespec)	import historical data by vintage dates
nrobs	import only new and revised observations
initial	import only first value for each observation in a series
long	import data in long format
<u>nosummary</u>	suppress summary table
clear	clear data in memory before importing FRED series

* `serieslist()` is required if `series_list` is not specified.
collect is allowed with `fredsearch`; see [U] 11.1.10 Prefix commands.
clear does not appear in the dialog box.

If start and end are provided as dates, they must be daily dates using notation of the form 31Jan2016, 2016-01-31, 2016/01/31, or 01/31/2016.

datespec may be

date	a daily date
date ₁ date ₂ ... date _n	a list of daily dates
_all	all available dates

<i>search_options</i>	Description
<code>idonly</code>	require <i>keywords</i> to appear in series IDs only
<code>tags(tag_list)</code>	search by <i>tag_list</i>
<code>taglist</code>	list tags present in current search results
<code>sort(sortby[, sortorder])</code>	list matched series in order specified by <i>sortby</i>
<code>detail</code>	list full metainformation for each search result
<code>saving(filename[, replace])</code>	save series information to <i>filename.dta</i>

`saving()` does not appear in the dialog box.

Options

Options are presented under the following headings:

[Option for set fredkey](#)
[Options for import fred](#)
[Options for freddescribe](#)
[Options for fredsearch](#)

Option for set fredkey

`permanently` specifies that, in addition to setting the key for the current Stata session, the key be remembered and become the default key when you invoke Stata.

Options for import fred

`serieslist(filename)` allows you to import the series specified in *filename*. The series file must contain a variable called `seriesid` that contains the IDs of the series you wish to import. `serieslist()` is required if *series_list* is not specified.

`daterange(start end)` specifies that only observations between the *start* date and *end* date should be imported. *start* and *end* must be specified as either a daily date or a missing value (`.`). Use `daterange(. end)` to import all observations from the first available through *end*. Use `daterange(start .)` to import from *start* through the most recently available date.

`aggregate(frequency[, method])` specifies that the data should be imported at a lower frequency than the series' native frequency along with an optional method of aggregation.

frequency may be `daily`, `weekly`, `biweekly`, `monthly`, `quarterly`, `semiannual`, `annual`, `weekly ending friday`, `weekly ending thursday`, `weekly ending wednesday`, `weekly ending tuesday`, `weekly ending monday`, `weekly ending sunday`, `weekly ending saturday`, `biweekly ending wednesday`, or `biweekly ending monday`.

method may be `avg` (the within-period average), `sum` (the within-period sum), or `eop` (the end-of-period value). The default is `avg`.

`realtime(start end)` specifies a real-time period between which all vintages for each series are imported. The vintage available on *start* is imported, as are all vintages released between *start* and *end*. Either of *start* or *end* may be replaced by a missing value (`.`). If *start* is a missing value, then all vintages from the first available up through *end* are imported. If *end* is a missing value, then all vintages from *start* up through the most recent available are imported. `realtime()` may not be combined with `vintage()`.

`vintage(datespec)` imports historical vintage data according to *datespec*. *datespec* may either be a list of daily dates or `_all`. When *datespec* is a list of dates, the specified series are imported as they were available on the dates in *datespec*. When *datespec* is `_all`, all vintages of the specified series are imported. `vintage()` may not be combined with `realtime()`.

`nrobs` specifies that only observations that are new or revised in each vintage be imported. Old and unrevised observations are imported as the missing value `.u`.

`initial` specifies that only the first value for each observation of the series be imported. This option may not be combined with `nrobs`.

`long` specifies that each series be imported in long format.

`nosummary` suppresses the summary table.

The following option is available with `import fred` but is not shown in the dialog box:

`clear` specifies that the data in memory should be replaced with the imported FRED data.

Options for `freddescribe`

`detail` displays full metainformation available about *series_list*.

`realtime(start end)` provides historical vintage information about *series_list* during the real-time period specified by *start* and *end*. Either *start* or *end* may be replaced by a missing value (`.`). If *start* is a missing value, then all vintages from the first available up through *end* are described. If *end* is a missing value, then all vintages from *start* up through the most recent available are described.

Options for `fredsearch`

`idonly` specifies that the keywords in *keyword_list* be found in series IDs rather than elsewhere in the metadata.

`tags(tag_list)` searches for series that have all the tags specified in *tag_list*. The complete list of available tags is provided by FRED. Tags form a space-separated list. Tags are case-sensitive and all FRED tags are in lowercase.

`taglist` lists all the tags present in the current search results.

`sort(sortby[, sortorder])` lists the search results in the order specified by *sortby*.

When searching series, *sortby* may be `popularity`, `id`, `title`, `lastupdated`, `frequency`, `obsstart`, `obsend`, `units`, or `seasonaladj`. By default, `popularity` is used.

When searching with the `taglist` option, *sortby* may be `name` or `series_count`. `name` means the tag name, and `series_count` is the count of series associated with the tag in the search results. By default, `series_count` is used.

You can optionally change the order of the search results from descending (`descending`) to ascending (`ascending`) order. The default order when searching by `popularity`, `lastupdated`, or `series_count` is descending; otherwise, the default sort order is ascending.

`detail` lists full metainformation for each series that appears in the search results.

The following option is available with `fredsearch` but is not shown in the dialog box:

`saving(filename[, replace])` saves the search results to a file. The filename may then be specified in the `serieslist()` option of `import fred` to import the series located by the search. The optional `replace` specifies that *filename* be overwritten if it exists.

Remarks and examples

Remarks are presented under the following headings:

Introduction and setup

The FRED interface

Advanced imports using the import fred command

Importing historical vintage data

Searching, saving, and retrieving series information

Describing series

Introduction and setup

`import fred` imports data from the Federal Reserve Economic Data (FRED) into Stata. FRED is maintained by the Economic Research Division of the Federal Reserve Bank of St. Louis and contains hundreds of thousands of economic and financial time series. FRED includes data from a variety of sources, including the Federal Reserve, the Penn World Table, Eurostat, the World Bank, and US statistical agencies, among others. `import fred` extends `freduse` discussed in [Drukker \(2006\)](#).

Series in FRED are updated and revised over time as new observations are added and as older observations are revised in light of more complete source information. The series are updated on an annual, quarterly, monthly, weekly, or daily basis, depending on the series. Each time a series is updated or revised, a new “vintage” is created. The archived data, or historical vintage data, are data in their unrevised form as they would have been available on a particular date in history. These data are from Archival FRED, or ALFRED. `import fred` can import data from either FRED or ALFRED.

FRED data can be imported using the `import fred` command or using the FRED interface. If you are exploring FRED, learning the names of series, or importing series occasionally, we recommend using the FRED interface. If you already know the names of the series that you would like to import or if you repeatedly download series as they are updated, we recommend using the `import fred` command. You may also use the FRED interface to learn series names that you subsequently specify in `import fred` commands. See [The FRED interface](#) below to learn more about using this tool.

Whether you plan to use the FRED interface or the `import fred` command, you must first have a valid API key. API keys are provided by the St. Louis Federal Reserve and may be obtained from https://research.stlouisfed.org/docs/api/api_key.html. The key will be a 32-character alphanumeric string. You will be prompted to enter this key the first time you open the FRED interface. Alternatively, you can type

```
. set fredkey key, permanently
```

where *key* is your API key.

► Example 1: A basic search and import

Suppose we want monthly data on the exchange rate between the US dollar and the Japanese Yen. We can use `fredsearch` to find the name of this series in FRED.

```
. fredsearch us dollar yen exchange rate monthly
```

Series ID	Title	Data range	Frequency
EXJPUS	Japanese Yen to U....	1971-01-01 to 2025-02-01	Monthly

```
Total: 1
```

The output says that EXJPUS is the name that FRED uses for this series. When we performed this search, 2025-02-01 was the last available observation. More data will be available when you type this command, so the endpoint of the data range will be more recent.

Having learned from the output that EXJPUS is the name that FRED uses for this series, we use `import fred` to import it.

```
. import fred EXJPUS
```

```
Summary
```

Series ID	Nobs	Date range	Frequency
EXJPUS	650	1971-01-01 to 2025-02-01	Monthly

```
# of series imported: 1
  highest frequency: Monthly
  lowest frequency: Monthly
```

The output says that 650 monthly observations on EXJPUS were imported.

To clarify what we imported, we can describe the imported data and list the first five observations.

```
. describe
```

```
Contains data
```

```
Observations:      650
Variables:         3
```

Variable name	Storage type	Display format	Value label	Variable label
datestr	str10	%-10s		observation date
daten	int	%td		numeric (daily) date
EXJPUS	float	%9.0g		Japanese Yen to U.S. Dollar Spot Exchange Rate

```
Sorted by: datestr
```

```
Note: Dataset has changed since last saved.
```

```
. list datestr daten EXJPUS in 1/5
```

	datestr	daten	EXJPUS
1.	1971-01-01	01jan1971	358.02
2.	1971-02-01	01feb1971	357.545
3.	1971-03-01	01mar1971	357.5187
4.	1971-04-01	01apr1971	357.5032
5.	1971-05-01	01may1971	357.413

Each series in FRED is paired with a string variable that records the daily date for each observation. `import fred` imports this daily date variable as the string variable `datestr`, and it creates `daten`, which is a Stata datetime variable that encodes the date in `datestr`. EXJPUS contains the observations on the FRED series EXJPUS.

Each series has metadata associated with it that is stored in the characteristics and may be viewed with the `char list` command. We now list out the metadata on EXJPUS.

```
. char list EXJPUS[]
EXJPUS[Title]:      Japanese Yen to U.S. Dollar Spot Exchange Rate
EXJPUS[Series_ID]:  EXJPUS
EXJPUS[Source]:     Board of Governors of the Federal Reserve Syste..
EXJPUS[Release]:    G.5 Foreign Exchange Rates
EXJPUS[Seasonal_Adjustment]:
                    Not Seasonally Adjusted
EXJPUS[Date_Range]: 1971-01-01 to 2025-02-01
EXJPUS[Frequency]:  Monthly
EXJPUS[Units]:      Japanese Yen to One U.S. Dollar
EXJPUS[Last_Updated]: 2025-03-03 15:19:41-06
EXJPUS[Notes]:      Averages of daily figures. Noon buying rates in..
```

See [P] `char` for more about characteristics.




The FRED interface

The names of FRED series are not predictable. The FRED interface makes it easy to find series, to import series, and to explore the thousands of series by keyword searches or by browsing by category, release type, source, or release date.

Selecting

File > Import > Federal Reserve Economic Data (FRED)

from the menu opens the FRED interface.




In the top left-hand corner, the drop-down menu defaults to Search FRED, which searches for series by keywords that appear in those series’ metadata. From this menu, we can also select Browse by category, Browse by release, Browse by source, and Search by release date.

- Browse by category finds series by browsing through FRED defined categories, such as Production & Business Activity.
- Browse by release finds series by browsing through FRED defined release types, such as the BEA Regions Employment and Unemployment and the Consumer Price Index.
- Browse by source finds series by browsing through sources, such as the Bank of England, the US Bureau of the Census, and the University of Pennsylvania.
- Search by release date finds regularly released series that were updated in a specified date range.

➤ Example 2: Finding and importing series with the FRED interface

Suppose we want to import series measuring the real gross domestic product (GDP) in the US and the three-month Treasury bill interest rate, known as the Federal Funds Rate. We can use a keyword search and a then browse by category to find and select them for import.

After selecting **File > Import > Federal Reserve Economic Data (FRED)** to open the control panel, we type `real gross domestic product us` in the `Keywords` field and click on the **Search** button, which produces



Clicking on **GDPC1** and then on the **Add** button adds **GDPC1** to list of series to import.

Import Federal Reserve Economic Data

Search FRED

Keywords:
real gross domestic product us
☒ Full text ☐ Series ID
Search

Tags:
Sources
Releases
Seasonal Adjustment
Frequencies
Geography Types
Geographies
Concepts
Add to filters

Filters:
Remove

Sort by: Popularity Descend

#	ID	Title
1	GDPC1	Real Gross Domestic Product
2	A939RX0Q048SB...	Real gross domestic product per capita
3	GDPPOT	Real Potential Gross Domestic Product
4	A191RL1Q225SB...	Real Gross Domestic Product
5	PCECC96	Real Personal Consumption Expenditures
6	GPDI1	Real Gross Private Domestic Investment
7	A261RX1Q020SB...	Real gross domestic income
8	STLENI	St. Louis Fed Economic News Index: Real GDP Nowcast
9	GDPC1CTM	FOMC Summary of Economic Projections for the Gro...
10	GDPCA	Real Gross Domestic Product
11	GCEC1	Real Government Consumption Expenditures and Gro...
12	PNFIC1	Real Private Nonresidential Fixed Investment
13	A261RL1Q225SB...	Real Gross Domestic Income
14	NYGDPPCAPKD...	Constant GDP per capita for the United States
15	A191RO1Q156N...	Real Gross Domestic Product
16	PRFIC1	Real Private Residential Fixed Investment
17	PB0000091Q225...	Real Average of GDP and GDI
18	A191RL1A225NB...	Real Gross Domestic Product
19	A794RX0Q048SB...	Real personal consumption expenditures per capita
20	ND000334Q	Real Gross Domestic Product
21	CPGDPAI	Contributions to Percent Change in Real GDP by Indu...
22	IMPWSC1	Real imports of goods and services
23	DPCERY2Q224S...	Contributions to percent change in real gross domest...
24	GDPC1MD	FOMC Summary of Economic Projections for the Gro...

1-1000 of 32992 Describe Add

Series to import:

#	Title	
1	Real Gross Domestic Product	GD

Remove Import Cancel

Ready

Now, we want to add the interest rate series. We select Browse by category from the drop-down menu in the top left-hand corner.

Import Federal Reserve Economic Data

Browse by category

Home > Categories

#	Categories
1	Money, Banking, & Finance
2	Population, Employment, & Labor Markets
3	National Accounts
4	Production & Business Activity
5	Prices
6	International Data
7	U.S. Regional Data
8	Academic Data

Tags:

Add to filters

Filters:

Remove

Sort by: Popularity

Descend

#	ID	Title
---	----	-------

Describe

Add

Series to import:

#	Title	GD
1	Real Gross Domestic Product	GD

Remove

Import

Cancel

Ready

We double-click on Money, Banking, & Finance to get a list of subcategories.

Import Federal Reserve Economic Data

Browse by category

Home > Categories > Money, Banking, & Finance

#	Money, Banking, & Finance
1	Interest Rates
2	Exchange Rates
3	Monetary Data
4	Financial Indicators
5	Banking
6	Business Lending
7	Foreign Exchange Intervention

Tags:

Add to filters

Filters:

Remove

Sort by: Popularity Descend

#	ID	Title
---	----	-------

< >

Describe Add

Series to import:

#	Title	GD
1	Real Gross Domestic Product	GD

< >

Remove Import Cancel

Ready

Next, we double-click on Interest Rates to get a list of interest-rate categories. Scrolling down, we find Treasury Bills.

Import Federal Reserve Economic Data

Browse by category

Home > Categories > Money, Banking, & Finance > Interest Rates

#	Interest Rates
16	Money Market Accounts
17	Mortgage Rates
18	Personal Loan Rates
19	Prime Bank Loan Rate
20	Saving Accounts
21	SONIA Rates
22	Treasury Bills
23	Treasury Constant Maturity
24	Treasury Inflation-Indexed Securities

Tags:

Sort by: Popularity Descend

Filters:

1-34 of 34

Describe Add

Series to import:

#	Title	GD
1	Real Gross Domestic Product	GD

Remove Import Cancel

Ready

We double-click on Treasury Bills to produce a list of interest-rate series. We click on TB3MS and then on the **Add** button to add it the list of series to be imported.

Import Federal Reserve Economic Data

Browse by category ▾

Home > Categories > Money, Banking, & Finance > Interest Rates > Treasury Bills

#	Categories
16	Money Market Accounts
17	Mortgage Rates
18	Personal Loan Rates
19	Prime Bank Loan Rate
20	Saving Accounts
21	SONIA Rates
22	Treasury Bills
23	Treasury Constant Maturity
24	Treasury Inflation-Indexed Securities

Tags:

- > Sources
- > Releases
- > Seasonal Adjustment
- > Frequencies
- > Geography Types
- > Geographies
- > Concepts

Add to filters

Filters:

Remove

Sort by: Popularity ▾ Descend ▾

#	ID	Title
1	TB3MS	3-Month Treasury Bill Secondary Market Rate, Discou...
2	DTB3	3-Month Treasury Bill Secondary Market Rate, Discou...
3	DTB1YR	1-Year Treasury Bill Secondary Market Rate, Discoun...
4	DTB4WK	4-Week Treasury Bill Secondary Market Rate, Discoun...
5	DTB6	6-Month Treasury Bill Secondary Market Rate, Discoun...
6	TB1YR	1-Year Treasury Bill Secondary Market Rate, Discoun...
7	WTB3MS	3-Month Treasury Bill Secondary Market Rate, Discoun...
8	TB4WK	4-Week Treasury Bill Secondary Market Rate, Discoun...
9	TB6MS	6-Month Treasury Bill Secondary Market Rate, Discoun...
10	RIFSGFSM03NA	3-Month Treasury Bill Secondary Market Rate, Discou...
11	WTB6MS	6-Month Treasury Bill Secondary Market Rate, Discou...
12	DTB1	1-Year Treasury Bill: Secondary Market Rate (DISCO...
13	TB3MA	3-Month Treasury Bill Rate: Auction Average (DISCO...
14	RIFSGFSY01NA	1-Year Treasury Bill Secondary Market Rate, Discoun...
15	WTB1YR	1-Year Treasury Bill Secondary Market Rate, Discoun...
16	WTB4WK	4-Week Treasury Bill Secondary Market Rate, Discou...

1-34 of 34

Describe Add

Series to import:

#	Title	
1	Real Gross Domestic Product	GD
2	3-Month Treasury Bill Secondary Market Rat...	TB

Remove Import Cancel

Ready

Clicking on import brings up a dialog box that allows us to restrict the imported observations.

We click **OK** to import all available observations.

The output from the command issued by the control panel was

```
. import fred GDPC1 TB3MS
```

Summary

Series ID	Nobs	Date range	Frequency
GDPC1	312	1947-01-01 to 2024-10-01	Quarterly
TB3MS	1094	1934-01-01 to 2025-02-01	Monthly

```
# of series imported: 2
highest frequency: Monthly
lowest frequency: Quarterly
```

The number of observations and the date ranges will differ when you follow these same steps using the FRED interface, because more data have been made available.



➤ Example 3: Refining a search using tags

Suppose that we want to find and import data on the median income in each US state and the District of Columbia for each available year. After opening the control panel, typing median household income in the Keywords box, and clicking on the **Search** button, we see

Import Federal Reserve Economic Data

Search FRED

Keywords:
median household income

Full text

Series ID

Tags:

- Sources
- Releases
- Seasonal Adjustment
- Frequencies
- Geography Types
- Geographies
- Concepts

Add to filters

Filters:

Remove

Sort by: Popularity Descend

#	ID	Title
1	MEHOINUSA672N	Real Median Household Income in the United States
2	MEHOINUSA646N	Median Household Income in the United States
3	MEHOINUSCA6...	Median Household Income in California
4	MHICA06037A05...	Estimate of Median Household Income for Los Angeles Co...
5	MEHOINUSNYA6...	Real Median Household Income in New York
6	MEHOINUSCOA6...	Real Median Household Income in Colorado
7	MEHOINUSMNA6...	Real Median Household Income in Minnesota
8	MHICA06075A05...	Estimate of Median Household Income for San Francisco C...
9	MEHOINUSTXA67...	Real Median Household Income in Texas
10	MEHOINUSWIA67...	Real Median Household Income in Wisconsin
11	MEHOINUSILA672N	Real Median Household Income in Illinois
12	MEHOINUSMIA672N	Real Median Household Income in Michigan
13	MEHOINUSFLA67...	Real Median Household Income in Florida
14	MEHOINUSNCA67...	Real Median Household Income in North Carolina
15	MEHOINUSMOA6...	Real Median Household Income in Missouri
16	MEHOINUSMAA6...	Real Median Household Income in Massachusetts
17	MEHOINUSCAA6...	Real Median Household Income in California
18	MHICA06073A05...	Estimate of Median Household Income for San Diego Count...
19	MEHOINUSOHA6...	Real Median Household Income in Ohio
20	MEHOINUSAZA6...	Real Median Household Income in Arizona
21	MEHOINUSVA6...	Real Median Household Income in Virginia
22	MHICA06059A05...	Estimate of Median Household Income for Orange County, ...

1-1000 of 9710

Series to import:

#	Title
---	-------

Remove Import Cancel

Ready

This keyword search finds thousands more series than the 51 we want. To filter the found series by the tag `state`, we expand the Geography Types category, click on `state`, and then click on the **Add to filters** button, which produces

Import Federal Reserve Economic Data

Search FRED

Keywords:
median household income

Full text Series ID

Tags:

- Sources
- Releases
- Seasonal Adjustment
- Frequencies
- Geographies
- Concepts

Add to filters

Filters:
state

Remove

Sort by: Popularity Descend


#	ID	Title
1	MEHOINUSCAA6...	Median Household Income in California
2	MEHOINUSNYA6...	Real Median Household Income in New York
3	MEHOINUSCOA6...	Real Median Household Income in Colorado
4	MEHOINUSMNA6...	Real Median Household Income in Minnesota
5	MEHOINUSTXA67...	Real Median Household Income in Texas
6	MEHOINUSWIA67...	Real Median Household Income in Wisconsin
7	MEHOINUSILA672N	Real Median Household Income in Illinois
8	MEHOINUSMIA672N	Real Median Household Income in Michigan
9	MEHOINUSFLA67...	Real Median Household Income in Florida
10	MEHOINUSNCA67...	Real Median Household Income in North Carolina
11	MEHOINUSMOA6...	Real Median Household Income in Missouri
12	MEHOINUSMAA6...	Real Median Household Income in Massachusetts
13	MEHOINUSCAA6...	Real Median Household Income in California
14	MEHOINUSOHA6...	Real Median Household Income in Ohio
15	MEHOINUSAZA6...	Real Median Household Income in Arizona
16	MEHOINUSVAA6...	Real Median Household Income in Virginia
17	MEHOINUSKYA6...	Real Median Household Income in Kentucky
18	MEHOINUSINA672N	Real Median Household Income in Indiana
19	MEHOINUSWVA6...	Real Median Household Income in West Virginia
20	MEHOINUSNJA67...	Real Median Household Income in New Jersey
21	MEHOINUSNVA6...	Median Household Income in Nevada
22	MEHOINUSLAA64...	Median Household Income in Louisiana

Series to import:

#	Title
---	-------

Ready

There are still too many series. To filter the series by the tag `real`, we expand the Concepts category, click on `real`, and then click on the **Add to filters** button, which produces the desired 51 series.



After selecting the 51 series, we add them to the import list by clicking on the **Add** button. We could now import them by clicking on the **Import** button.



Advanced imports using the import fred command

FRED data users commonly import series of different frequencies.

► Example 4: Importing series with different frequencies

Suppose we wish to import current data on US real GDP, the price level, and the interest rate. These data are stored in FRED with the series IDs “GDPC1”, “GDPDEF”, and “FEDFUNDS”, so we supply those names to import fred.

```
. import fred GDPC1 GDPDEF FEDFUNDS
```

Summary

Series ID	Nobs	Date range	Frequency
GDPC1	312	1947-01-01 to 2024-10-01	Quarterly
GDPDEF	312	1947-01-01 to 2024-10-01	Quarterly
FEDFUNDS	848	1954-07-01 to 2025-02-01	Monthly

```
# of series imported: 3
```

```
highest frequency: Monthly
```

```
lowest frequency: Quarterly
```

FEDFUNDS is a monthly series, while GDPC1 and GDPDEF are quarterly series. To further illustrate, we list the observations on each variable from 1959 using the `list` command.

```
. list if year(daten)==1959, separator(3)
```

	datestr	daten	GDPC1	GDPDEF	FEDFUNDS
85.	1959-01-01	01jan1959	3352.129	15.224	2.48
86.	1959-02-01	01feb1959	.	.	2.43
87.	1959-03-01	01mar1959	.	.	2.8
88.	1959-04-01	01apr1959	3427.667	15.248	2.96
89.	1959-05-01	01may1959	.	.	2.9
90.	1959-06-01	01jun1959	.	.	3.39
91.	1959-07-01	01jul1959	3430.057	15.307	3.47
92.	1959-08-01	01aug1959	.	.	3.5
93.	1959-09-01	01sep1959	.	.	3.76
94.	1959-10-01	01oct1959	3439.832	15.367	3.98
95.	1959-11-01	01nov1959	.	.	4
96.	1959-12-01	01dec1959	.	.	3.99

FRED provides all series in daily date format, and each observation is recorded as existing on the first day of the period. For example, a monthly series records the observation in 1959 January as existing on 01Jan1959; a quarterly series records the observation in 1959 Q1 as existing on 01Jan1959. When importing series of different frequencies, the lower-frequency series will appear to contain gaps; these gaps are filled with missing values.

◀

► Example 5: Importing series at a desired frequency

Continuing with [example 4](#), at times you may wish to import a high-frequency series at a particular lower frequency. This is accomplished with the `aggregate()` option. There are three aggregation methods available: you may take the within-period average, the sum, or the end-of-period value. The default is to take the within-period average.

```
. import fred GDPC1 GDPDEF FEDFUNDS, aggregate(quarterly) clear
```

Summary

Series ID	Nobs	Date range	Frequency
GDPC1	312	1947-01-01 to 2024-10-01	Quarterly
GDPDEF	312	1947-01-01 to 2024-10-01	Quarterly
FEDFUNDS	282	1954-07-01 to 2024-10-01	Quarterly

```
# of series imported: 3
  highest frequency: Quarterly
  lowest frequency: Quarterly
. list if year(daten)==1959, separator(4)
```

	datestr	daten	GDPC1	GDPDEF	FEDFUNDS
49.	1959-01-01	01jan1959	3352.129	15.224	2.57
50.	1959-04-01	01apr1959	3427.667	15.248	3.08
51.	1959-07-01	01jul1959	3430.057	15.307	3.58
52.	1959-10-01	01oct1959	3439.832	15.367	3.99

The monthly series FEDFUNDS has been reduced to quarterly frequency. The value of FEDFUNDS for the first quarter of 1959, 2.57, is the average of its values for the three months in that quarter. The date variable `daten` now stores the first date of each quarter.



► Example 6: Importing a subset of observations

The `daterange()` option causes `import fred` to restrict importing of data to only observations within the specified beginning and ending dates. `daterange()` takes two arguments, both of which must be either daily dates or missing (`.`). If a missing value is used for the first date, then all observations from the beginning up to the end date are imported. If a missing value is used for the second date, then all observations from the first date through the most current are imported.

Returning to [example 4](#), we may wish to import only data between 1984 and 2005 for `GDPC1`, `GDPDEF`, and `FEDFUNDS`.

```
. import fred GDPC1 GDPDEF FEDFUNDS, daterange(1984-01-15 2005-12-31) clear
```

Summary

Series ID	Nobs	Date range	Frequency
GDPC1	88	1984-01-01 to 2005-10-01	Quarterly
GDPDEF	88	1984-01-01 to 2005-10-01	Quarterly
FEDFUNDS	264	1984-01-01 to 2005-12-01	Monthly

```
# of series imported: 3
```

```
highest frequency: Monthly
```

```
lowest frequency: Quarterly
```

Note that `GDPC1` and `GDPDEF` now have 88 observations rather than 278; similarly, `FEDFUNDS` has 264 observations rather than 745.



Importing historical vintage data

In [example 1](#), we imported monthly data on the exchange rate between the US Dollar and the Japanese Yen. The observations on `EXJPUS` listed in that example were observed end-of-day values. In contrast, the values in many FRED series, like the US real gross domestic product series (`GDPC1`), are estimates. The values of observed series do not change over time. The values of estimated series change over time because the rules that define them change over time. A set of rules is known as a vintage.

FRED contains the most recent vintage of a given series. At times, you may wish to import prior vintages or to view the series as it would have been seen on a particular date in history. ALFRED contains prior vintages of economic data and allows you to import data as they were seen on a particular date in history. For example, you may import the real GDP series that you would have had access to on October 15, 2008.

By default, `import fred` imports data from the current vintage. The `vintage()` and `realtime()` options allow you to import data from prior vintages. You can request a single date, multiple dates, all vintages between two dates in history, or the complete revision history.

► Example 7: Importing vintages by date

We wish to import the gross national product (GNP) series as it would have been available on September 16, 2008 and September 16, 2009, so we specify these dates in the `vintage()` option. We also use the `daterange()` option to import only observations since 2006:

```
. import fred GNPC96, vintage(2008-09-16 2009-09-16) daterange(2006-01-01 .)
> clear
```

Summary

Series ID	Nobs	Date range	Frequency
GNPC96_20080916	10	2006-01-01 to 2008-04-01	Quarterly
GNPC96_20090916	14	2006-01-01 to 2009-04-01	Quarterly

```
# of series imported: 2
highest frequency: Quarterly
lowest frequency: Quarterly
. list, separator(4) abbreviate(16)
```

	datestr	daten	GNPC96_20080916	GNPC96_20090916
1.	2006-01-01	01jan2006	11286.5	12994.2
2.	2006-04-01	01apr2006	11365.1	13035.4
3.	2006-07-01	01jul2006	11370.8	13025.1
4.	2006-10-01	01oct2006	11426.5	13129.5
5.	2007-01-01	01jan2007	11419.1	13160.5
6.	2007-04-01	01apr2007	11541.7	13275.9
7.	2007-07-01	01jul2007	11719.9	13451.5
8.	2007-10-01	01oct2007	11758.3	13563.3
9.	2008-01-01	01jan2008	11760.9	13525.4
10.	2008-04-01	01apr2008	11835.9	13533.7
11.	2008-07-01	01jul2008	.	13470.7
12.	2008-10-01	01oct2008	.	13240.5
13.	2009-01-01	01jan2009	.	13018.1
14.	2009-04-01	01apr2009	.	12991.6

We specified one series and two vintage dates, so we have imported two series. Each vintage is named with the series requested and the date that it was requested. For example, the series `GNPC96_20080916` reports real GNP as it was available on 16 September 2008. Note that the series is appended with the date requested, not the date the vintage was released.

These two vintages of `GNPC96` differ dramatically because they are on different scales. The output also illustrates that, as of 16 September 2008, data on `GNPC96` were only available through 1 April 2008.

◀

► Example 8: Importing vintages by real-time period

You may also wish to obtain the complete vintage history of a series between two dates. For example, we import all the vintages of real GNP from December 2007 through July 2010 by specifying this date range in the `realtime()` option.

```
. import fred GNPC96, realtime(2007-12-01 2010-07-31) clear
```

```
Summary
```

Series ID	Nobs	Date range	Frequency
GNPC96_20071201	243	1947-01-01 to 2007-07-01	Quarterly
GNPC96_20071220	243	1947-01-01 to 2007-07-01	Quarterly
GNPC96_20080327	244	1947-01-01 to 2007-10-01	Quarterly
GNPC96_20080529	245	1947-01-01 to 2008-01-01	Quarterly
GNPC96_20080626	245	1947-01-01 to 2008-01-01	Quarterly
GNPC96_20080731	245	1947-01-01 to 2008-01-01	Quarterly
GNPC96_20080828	246	1947-01-01 to 2008-04-01	Quarterly
GNPC96_20080926	246	1947-01-01 to 2008-04-01	Quarterly
GNPC96_20081125	247	1947-01-01 to 2008-07-01	Quarterly
GNPC96_20081223	247	1947-01-01 to 2008-07-01	Quarterly
GNPC96_20090326	248	1947-01-01 to 2008-10-01	Quarterly
GNPC96_20090529	249	1947-01-01 to 2009-01-01	Quarterly
GNPC96_20090625	249	1947-01-01 to 2009-01-01	Quarterly
GNPC96_20090731	249	1947-01-01 to 2009-01-01	Quarterly
GNPC96_20090817	249	1947-01-01 to 2009-01-01	Quarterly
GNPC96_20090827	250	1947-01-01 to 2009-04-01	Quarterly
GNPC96_20090930	250	1947-01-01 to 2009-04-01	Quarterly
GNPC96_20091124	251	1947-01-01 to 2009-07-01	Quarterly
GNPC96_20091222	251	1947-01-01 to 2009-07-01	Quarterly
GNPC96_20100326	252	1947-01-01 to 2009-10-01	Quarterly
GNPC96_20100527	253	1947-01-01 to 2010-01-01	Quarterly
GNPC96_20100625	253	1947-01-01 to 2010-01-01	Quarterly
GNPC96_20100730	253	1947-01-01 to 2010-01-01	Quarterly
GNPC96_20100731	253	1947-01-01 to 2010-01-01	Quarterly

```
# of series imported: 24
```

```
highest frequency: Quarterly
```

```
lowest frequency: Quarterly
```

Each series contains the data from a vintage, and each series' name is appended with the date that the vintage was released.



Different vintages of a series may not be directly comparable. For example, the units of a series may change over time. The different vintages must be converted to a common unit before they are analyzed, and it is crucial that you be aware of the units of the vintages you are analyzing.

Note that there is slightly different behavior depending on whether you specify vintage dates or import all vintages within a real-time period. If you specify a list of dates, then each vintage will be named `series_date`. On the other hand, if you import every vintage between two dates using the `realtime()` option, then each vintage will be named `series_vintage_date`. This behavior follows FRED's behavior when handling vintages.

Searching, saving, and retrieving series information

`fredsearch` finds series that match keywords or tags. Around 5,000 tags are supplied by FRED. You can also search by keywords, which will search for the keyword anywhere in the metadata of a series.

You can save the names of the series found by a search to a file and then import these series. The following example uses tags in combination with keywords to import median income per capita for states in the United States.

➤ Example 9: Using the search engine

Suppose we wish to import median income per capita for each state. This requires us to identify 51 series, one for each state and the District of Columbia. The series IDs may follow some pattern, but it is not immediately obvious what those IDs are. We could use the FRED interface, as in [example 3](#), or we could use `fredsearch` to search for the relevant series, save the IDs to a file, and use that file to load the correct series. This example takes the latter approach.

The `fredsearch` command invokes the search engine. `fredsearch keywords` allows you to search for *keywords* anywhere in the series metadata. The `tags()` option allows you to filter the search results using some of FRED’s 5,000 designated tags.

```
. fredsearch median household income, tags(state real)
```

Series ID	Title	Data range	Frequency
MEHOINUSNYA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSCAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSTXA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSMIA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSFLA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSMNA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSDCA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSAZA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSMAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSUTA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSPAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSALA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSOHA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSINA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSILA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSNJA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSCTA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSCOA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSVAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSKYA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSOKA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSNCA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSMOA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSWIA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSRIA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSWAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSHIA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSGAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSMSA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSARA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSIAA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSWVA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSSCA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSNEA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
MEHOINUSNHA672N	Real Median Househ...	1984-01-01 to 2023-01-01	Annual
(output omitted)			

Total: 51

In the above search command, we searched FRED for all series containing “median”, “household”, and “income” somewhere in their metadata, and restricted the search to series with the tags “state” (for states) and “real” (for inflation-adjusted series). The result is 51 series, one for each state and the District of Columbia.

fredsearch provides information about series but does not import them. We can save the search results to a file, then import all series that matched our search results:

```
. fredsearch median household income, tags(state real) saving(myfile.dta)
(51 series added to myfile.dta)

. import fred, serieslist(myfile.dta) clear

Summary
```

Series ID	Nobs	Date range	Frequency
MEHOINUSNYA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSCAA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSTXA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSMIA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSFLA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSMNA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSDCA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSAZA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSMAA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSUTA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSPAA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSALA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSOHA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSINA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSILA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSNJA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSCTA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSCOA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSVAA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSKYA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSOKA672N	40	1984-01-01 to 2023-01-01	Annual
MEHOINUSNCA672N	40	1984-01-01 to 2023-01-01	Annual
(output omitted)			

```
# of series imported: 51
highest frequency: Annual
lowest frequency: Annual
```

This example showed how to quickly import 51 series for median household income by state. A similar procedure can quickly isolate and import the roughly 200 series that report data on infant mortality by country or the roughly 200 series that report the investment share of GDP by country.



Describing series

freddescribe provides facilities to describe series based on their metadata. `freddescribe series_list` provides a brief summary of `series_list`. The series are only described, not imported.

With the `detail` option, detailed series metadata are displayed, including the full title of the series, the source agency, the source data release, seasonal adjustment, date range for which observations exist, frequency of observations, units, date and time that the series was last updated, and notes, which contain FRED’s notes about the series. Finally, the full metadata includes a list of all vintage dates associated with the series.

Specifying the `realtime(start end)` option on `freddescribe` provides information about a series by a real-time period. This option allows you to see how a series’ units have changed over time. `freddescribe` will display the series description for each vintage between the specified start and end dates.

`freddescribe, realtime(. end)` describes all vintages from the first available vintage up to that of *end*. Similarly, `freddescribe, realtime(start .)` describes all vintages from *start* up through the most current vintage available.

► Example 10: Describing series

Suppose we wish to know what vintages are available for real GDP, whose FRED series name is `GDPC1`. We use `freddescribe` with the `detail` option to list all the vintages.

```
. freddescribe GDPC1, detail
```

GDPC1

Title:

Source:

Release:

Seasonal adjustment:

Date range:

Frequency:

Units:

Last updated:

Notes:

Vintage dates:

Real Gross Domestic Product

U.S. Bureau of Economic Analysis

Gross Domestic Product

Seasonally Adjusted Annual Rate

1947-01-01 to 2024-10-01

Quarterly

Billions of Chained 2017 Dollars

2025-03-27 08:03:25-05

BEA Account Code: A191RX Real gross domestic product i...

1991-12-04 1991-12-20 1992-01-29 1992-02-28 1992-03-26

1992-04-28 1992-05-29 1992-06-25 1992-07-30 1992-08-27

1992-09-24 1992-10-27 1992-11-25 1992-12-22 1993-01-28

1993-02-26 1993-03-26 1993-04-29 1993-05-28 1993-06-23

1993-07-29 1993-08-31 1993-09-29 1993-10-28 1993-12-01

1993-12-22 1994-01-28 1994-03-01 1994-03-31 1994-04-28

1994-05-27 1994-06-29 1994-07-29 1994-08-26 1994-09-29

1994-10-28 1994-11-30 1994-12-22 1995-01-27 1995-03-01

1995-03-31 1995-04-28 1995-05-31 1995-06-30 1995-07-28

1995-08-30 1995-09-29 1995-10-27 1996-01-19 1996-02-23

(output omitted)

Total: 1

Vintages since 1991 are available for download. If we had not specified `detail`, only the series name, start and end date, and frequency would have been displayed.



► Example 11: Obtaining historical descriptions

Information for real GNP in the United States is contained in FRED series GNPC96. Real GNP is expressed in the units of some base year, and over time the base year changes. In this example, we will examine how the units for GNPC96 have changed over time by requesting a description of all vintages up through December 31, 2015 using the `realtime()` option.

```
. freddescribe GNPC96, realtime(. 2015-12-31)
```

Series ID	Real time	Units
GNPC96	1958-12-21 to 1959-02-18	Billions of 1957 Dollars
GNPC96	1959-02-19 to 1965-08-18	Billions of 1954 Dollars
GNPC96	1965-08-19 to 1976-01-15	Billions of 1958 Dollars
GNPC96	1976-01-16 to 1985-12-19	Billions of 1972 Dollars
GNPC96	1985-12-20 to 1991-12-03	Billions of 1982 Dollars
GNPC96	1991-12-04 to 1996-01-18	Billions of 1987 Dollars
GNPC96	1996-01-19 to 1999-10-28	Billions of Chained 1992 Dollars
GNPC96	1999-10-29 to 2003-12-09	Billions of Chained 1996 Dollars
GNPC96	2003-12-10 to 2009-07-30	Billions of Chained 2000 Dollars
GNPC96	2009-07-31 to 2013-07-30	Billions of Chained 2005 Dollars
GNPC96	2013-07-31 to 2015-12-31	Billions of Chained 2009 Dollars

Total: 11

Vintages for this series begin in 1958. A new row signifies a change in units. There are 11 total changes in units in GNPC96. Every vintage of GNPC96 between 2009-07-31 and 2013-07-30, for example, is in the units “Billions of chained 2005 dollars”. Meanwhile, vintages since 2013-07-30 are in units “Billions of chained 2009 dollars”. Real GNP vintages from 2010 and 2014 will not be immediately comparable due to the difference in units; they should be converted into a common unit before analysis.

Additional information by real-time period can be obtained by specifying the detail option. We can inspect the details of vintages since 2008:

```
. freddescribe GNPC96, detail realtime(2007-12-31 2013-01-15)
```

GNPC96	2007-12-31 to 2009-07-30
Title:	Real Gross National Product
Source:	U.S. Bureau of Economic Analysis
Release:	Gross Domestic Product
Seasonal adjustment:	Seasonally Adjusted Annual Rate
Date range:	1947-01-01 to 2009-01-01
Frequency:	Quarterly
Units:	Billions of Chained 2000 Dollars
Last updated:	2009-06-25 10:47:06-05
Notes:	BEA Account Code: A001RX1 A Guide to the National Inco...
Vintage dates:	2008-03-27 2008-05-29 2008-06-26 2008-07-31 2008-08-28 2008-09-26 2008-11-25 2008-12-23 2009-03-26 2009-05-29 2009-06-25

GNPC96	2009-07-31 to 2013-01-15
Title:	Real Gross National Product
Source:	U.S. Bureau of Economic Analysis
Release:	Gross Domestic Product
Seasonal adjustment:	Seasonally Adjusted Annual Rate
Date range:	1947-01-01 to 2012-07-01
Frequency:	Quarterly
Units:	Billions of Chained 2005 Dollars
Last updated:	2012-12-20 08:17:16-06
Notes:	BEA Account Code: A001RX1 A Guide to the National Inco...
Vintage dates:	2009-07-31 2009-08-17 2009-08-27 2009-09-30 2009-11-24 2009-12-22 2010-03-26 2010-05-27 2010-06-25 2010-07-30 2010-08-27 2010-09-30 2010-11-23 2010-12-22 2011-03-25 2011-05-26 2011-06-24 2011-07-29 2011-08-26 2011-09-29 2011-11-22 2011-12-22 2012-03-29 2012-05-31 2012-06-28 2012-07-27 2012-08-29 2012-09-27 2012-11-29 2012-12-20

Total: 2

The detail option provides much of the same information as it did without `realtime()`, but now a new detail block is provided for each vintage where the details themselves change. Most of the details remain constant across vintages, but in this example, “Units” and “Date range” are different for each block.

The vintage list is now separated, with each vintage falling into the appropriate describe block. For example, all vintages of GNPC96 in 2010 have metainformation corresponding to the block that describes vintages from 2009-07-31 to 2013-01-15.



Stored results

```
fredsearch stores the following in r():  
Scalars  
  r(series_ids)      list of series IDs contained in the search results
```

References

Drukker, D. M. 2006. [Importing Federal Reserve economic data](#). *Stata Journal* 6: 384–386.

Schenck, D. 2017. Importing data with import fred. *The Stata Blog: Not Elsewhere Classified*. <https://blog.stata.com/2017/08/08/importing-data-with-import-fred/>.

Also see

[D] [import](#) — Overview of importing data into Stata

[D] [import delimited](#) — Import and export delimited text data

[D] [import haver](#) — Import data from Haver Analytics databases

[D] [odbc](#) — Load, write, or view data from ODBC sources

[TS] [tsset](#) — Declare data to be time-series data

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

